

collect unget — Remove results from a collection

[Description](#)
[Options](#)
[Also see](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Stored results](#)

[Syntax](#)
[Reference](#)

Description

`collect unget` removes results from a collection.

Quick start

Remove coefficients and standard errors identified with omitted predictor variable `x1`

```
collect unget _r_b _r_se, fortags(colname[o.x1])
```

Remove all frequency results

```
collect unget frequency
```

Remove frequency results that are tagged with any level of dimension `group`

```
collect unget frequency, fortags(group)
```

Remove frequency results that are tagged with level `_hide` of dimension `group`

```
collect unget frequency, fortags(group[_hide])
```

Remove frequency results that are tagged with `code1[first]` or `code2[last]`

```
collect unget frequency, fortags(code1[first] code2[last])
```

Remove frequency results that are tagged with `code1[first]` and `code2[last]`

```
collect unget frequency, fortags(code1[first]#code2[last])
```

Menu

Statistics > Summaries, tables, and tests > Tables and collections > Remove results

Syntax

```
collect unget results [ , options ]
```

results is a list of one or more `result` levels that identify values to be removed from the collection.

<i>options</i>	Description
Main	
<code>name</code> (<i>cname</i>)	remove results from collection <i>cname</i>
Options	
<code>fortags</code> (<i>taglist</i>)	only add tags to values with tags in <i>taglist</i>

Options

Main

`name` (*cname*) specifies the collection from which to remove results. If this option is not specified, the change is made in the current collection.

Options

`fortags` (*taglist*) specifies conditions for selecting the results to be removed from the collection. Results with tags in *taglist* will be removed from the collection.

Within the *taglist*, if tags are joined by #, results having all of these tags are selected; if tags are separated by a space, results with any of these tags are selected.

taglist contains

tagspec

tagspec taglist

tagspec contains

tag

tag#tag [*#tag* [. . .]]

tag contains

dimension

dimension [*levels*]

dimension is a dimension in the collection.

levels are levels of the corresponding dimension.

Distinguish between [], which are to be typed, and [], which indicate optional arguments.

Remarks and examples

After collecting results, we may want to remove or replace some results to accommodate special requirements for the table we wish to create. `collect unget` allows you to remove results from a collection.

One example is a coefficient table with factor-variable coefficients, where we want to identify base levels with the string “(base)” in the column of standard errors.

To demonstrate, we use data from the Second National Health and Nutrition Examination Survey (NHANES II) (McDowell et al. 1981). We fit a model for systolic blood pressure, then arrange the fitted coefficients and standard errors into a table.

```
. use https://www.stata-press.com/data/r19/nhanes21
(Second National Health and Nutrition Examination Survey)
. quietly: collect: regress bpsystol weight i.diabetes i.sex
. collect layout (colname) (result[_r_b _r_se])
Collection: default
  Rows: colname
  Columns: result[_r_b _r_se]
Table 1: 6 x 2
```

	Coefficient	Std. error
Weight (kg)	.4340474	.0153533
Not diabetic	0	0
Diabetic	14.34115	1.019611
Male	0	0
Female	1.107633	.4710559
Intercept	98.40567	1.235476

We want to replace the zero-valued standard errors with the string “(base)”. First, we use `collect unget` to remove the standard error results for the base levels of `diabetes` and `sex`. Then, we use `collect preview` to verify our intended change.

```
. collect unget _r_se , fortags(colname[0.diabetes 1.sex])
(2 items removed from collection default)
. collect preview
```

	Coefficient	Std. error
Weight (kg)	.4340474	.0153533
Not diabetic	0	
Diabetic	14.34115	1.019611
Male	0	
Female	1.107633	.4710559
Intercept	98.40567	1.235476

Now let's add some string results to replace the zero values we just removed.

```
. collect get _r_se="(base)", tag(colname[0.diabetes])
. collect get _r_se="(base)", tag(colname[1.sex])
. collect preview
```

	Coefficient	Std. error
Weight (kg)	.4340474	.0153533
Not diabetic	0	(base)
Diabetic	14.34115	1.019611
Male	0	(base)
Female	1.107633	.4710559
Intercept	98.40567	1.235476

Stored results

collect unget stores the following in `s()`:

Macros

```
s(collection)  name of collection
s(k_removed)   number of items removed
```

Reference

McDowell, A., A. Engel, J. T. Massey, and K. Maurer. 1981. "Plan and operation of the Second National Health and Nutrition Examination Survey, 1976–1980". In *Vital and Health Statistics*, ser. 1, no. 15. Hyattsville, MD: National Center for Health Statistics.

Also see

[TABLES] [collect get](#) — Collect results from a Stata command

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

